
1 CES functions and the calibrated share form:  List of symbols

α : Distribution parameter
θ : Cost share
ρ : CES elasticity parameter
σ : Elasticity of substitution
A : Scaling factor
C : Cost function
K : Input of capital
L : Input of labor
P : Prices of capital, labor, output
X : Output

:  Initial observed values of variables



Textbook form of production function:

      

Calibrated share form of production function

where   remain fixed after calibration



Textbook for of cost function:

Calibrated share form:

Notational shortcut: since the producer price  equals marginal cost
in equilibrium, we can use  in place of marginal cost

 could be problems if X is slack



Textbook form of factor demand function:

Calibrated share form:

N.B. if the price of X goes up holding the price of L constant, it means
that the price of K is going up!



As prices change, the shares of L and K in output change

  

where is the initial (fixed) calibrated share



2. The MPS/GE subsystem of GAMS

GAMS includes a higher-level language, written by Rutherford, MPS/GE:
mathematical programming system for general equilibrium.  MPS/GE
uses the MCP solver in GAMS. 

First, the program has routines for calibrating and writing all constant-
returns CES and CET functions, up to 7 (?) levels of nesting. 

Second, and closely related, the form of the data required to specify a
CES/CET function is exactly the data modelers have, swift and easy
move from an accounting matrix to the calibration of the model.

Third, a lot of market-clearing and income-balance equations are written
automatically by MPS/GE so the modeler doesn’t have to. 

Fourth, and closely related, a lot of errors that can occur when a modeler
writes out his or her equations cannot occur in MPS/GE.  

If there is a tax or markup, for example, the revenues must be assigned to
some agent and will be allocated automatically to that agent by the
income-balance properties of the coding.

A few key words.

SECTOR  (ACTIVITY)

Production activities that convert commodity inputs into commodity
outputs.  The variable associated with a sector is the activity (output)
level.

For every sector/acivity, mps/ge constructs a cost function and zero-profit
inequality.

COMMODITY  (MARKETS)

A good or factor.  The variable associated with a commodity is its price,
not its quantity.

For every commodity/market, mps/ge constructs a market-clearing
inequality, using Shepard’s lemma for good / factor demands

CONSUMERS

Individuals who supply factors and receive tax revenues, markups, and
pay subsidies.  In imperfectly competitive models, firm owners can be
designated as consumers.  

A government that receives tax revenue and buys public goods is also
designated as a consumer.   The variable associated with a consumer
is income from all sources.

AUXILIARY

Additional variables, such as markup formulae or taxes with endogenous
values which are functions of other variables such as prices and
quantities.  Please note the spelling of auxiliary.

CONSTRAINT

An equation that is typically used to set the value of an auxiliary variable.
Constraint equations will be used to set the values of markups, which
are auxiliary variables, set quota values, etc.

Here is what an MPS/GE program, embedded in a GAMS file, looks like,
where the model name is kyiv2-mps

GAMS statements such as declaring sets, parameters, parameter values,
etc.

**** now control is passed to the MPS/GE subsystem ****

$ONTEXT [this tells the GAMS compiler to ignore what
follows, but the MPS/GE compiler will recognize the
model statement that follows and will begin to pay
attention]

$MODEL:KYIV2

Declare sectors, commodities, consumers, auxiliary
 variables

Production Blocks

Demand Blocks

Constraint equations

$OFFTEXT [control is passed back to GAMS]

**** now we are back in GAMS ****

$SYSINCLUDE MPSGESET KYIV2

GAMS statements such as setting starting values of
variables, other parameter values, etc.

$INCLUDE KYIV2.GEN
SOLVE KYIV2 USING MCP;

GAMS statements processing output
________________________________________________

Below, we formulate problem introduced above as M3-4b using MPS/GE

$TITLE: KYIV2-MPS  MPS/GE VERSION

$ONTEXT
 extends model JGEA1 by adding a tax and a (iceberg) transactions
cost
 and showing the substantial difference between the two
               Production Sectors     Consumers
   Markets   |    X       Y       W       I
   ------------------------------------------------------
        PX   |  100            -100  |
        PY   |          100    -100  |
        PL   |  -25     -75          |   100
        PK   |  -75     -25          |   100
        PW   |                  200  |  -200
   ------------------------------------------------------
$OFFTEXT

PARAMETERS

 TR      ad-valorem tax rate for X sector inputs
 TC      iceberg transport (trade) cost on X
 LBAR    labor endowment
 KBAR    capital endowment;

LBAR = 100;
KBAR = 100;
TR = 0; TC = 0;

$ONTEXT
$MODEL:KYIV2

$SECTORS:
        X       ! Activity level for sector X
        Y       ! Activity level for sector Y
        W       ! Activity level for sector W (welfare index)

$COMMODITIES:
        PX      ! Price index for commodity X
        PY      ! Price index for commodity Y
        PL      ! Price index for primary factor L
        PK      ! Price index for primary factor K
        PW      ! Price index for welfare (expenditure function)

$CONSUMERS:
        CONS    ! Income level for consumer CONS

$PROD:X s:1
        O:PX  Q:(100/(1+TC))
        I:PL  Q:25   A:CONS T:TR
        I:PK  Q:75   A:CONS T:TR

$PROD:Y s:1
        O:PY  Q:100
        I:PL  Q:75
        I:PK  Q:25

$PROD:W s:1
        O:PW   Q:200
        I:PX   Q:100
        I:PY   Q:100

$DEMAND:CONS
        D:PW   Q:200
        E:PL   Q:LBAR
        E:PK   Q:KBAR

$OFFTEXT
$SYSINCLUDE mpsgeset KYIV2

PW.FX = 1;

$INCLUDE KYIV2.GEN
SOLVE KYIV2 USING MCP;

* SHOW HOW TO DO MULTIPLE SCENARIOS
* SHOW DIFFERENCE BETWEEN TARIFF AND TRADE COST OF EQUAL RATES

SETS T indexes 25 different gross cost levels         /T1*T25/
     J indexes 2 scenarios: 1 = tariff 2 = trade cost /J1*J2/;

PARAMETERS
 RATE(T)          net tax or trade cost rate (gross rate minus 1)
 WELFARE(T,J)     welfare normalized to equal 1 in benchmark
 RESULTS(T, *)    formats results in one table;

LOOP(J,
LOOP(T,

TC = 0; TR = 0;
RATE(T) = 0.05*ORD(T) - 0.05;
TR$(ORD(J) EQ 1) = RATE(T);
TC$(ORD(J) EQ 2) = RATE(T);

$INCLUDE KYIV2.GEN
SOLVE KYIV2 USING MCP;

WELFARE(T,J) = W.L;

);
);

RESULTS(T, "RATE") = RATE(T);
RESULTS(T, "WELTR") = WELFARE(T, "J1");
RESULTS(T, "WELTC") = WELFARE(T, "J2");

DISPLAY RESULTS;

* Write parameter RESULTS to an Excel file KYIV2.XLS,
* starting in Sheet1

Execute_Unload 'KYIV2.gdx' RESULTS
execute 'gdxxrw.exe KYIV2.gdx par=RESULTS rng=SHEET2!A3'

Relationship between MCP and MPS/GE 

MCP MPS/GE

Zero profit (pricing) inequality PROD block

complementary variable: quantity

Market clearing inequality Constructed from PROD block
O, I, and E fields 

complementary variable: price

Income balance equation DEMAND block

complementary variable: income

Extra equations/inequalities CONSTRAINT block

complementary variable: extra / auxiliary

Production blocks

The terminology here is a bit confusing, since MPS/GE takes the
information in a production block and generates a cost function, not a
production function.  

But the variable complementary with a production block (cost function) is
an activity level (pricing equation <=> quantity variable). 

$PROD:Y  s:1
        O:PY    Q:100   P:1
        I:PL    Q: 75   P:1
        I:PK    Q: 25   P:1

First line

Name of activity (Y), value of substitution (here s:1) and
transformation elasticities if there are several outputs.  

Default elasticity of substitution is 0 (not 1!).

First column

Names of commodity outputs (O:) and inputs (I:).

Second column

Reference commodity quantities (Q) – used for calibration.  Default =
1 if none specified.

Third column

Reference commodity prices (P) – used for calibration.  Default = 1 if
none specified, so they are omitted in the program above.

MPS/GE then takes this information to construct a cost function and,
as a feature of CES functions, it is globally defined by a single
reference point.  

Think of putting an isoquant labeled 100 units of output, with elasticity of
substitution 1, though input points L = 75, K = 25, with slope PL/PK =
1.  

With constant returns to scale, this information is sufficient to construct
the (global) cost function.   

In this simple case, it constructs the cost function and zero-profit
inequality:

100*(PL**.75)*(PK**.25) =G= 100*PY;

The saving from using MPS/GE might not seem like a big deal, but
believe me with many inputs, different prices for all inputs, and an
elasticity of substitution of 3.5, it is a huge saving indeed.



100*(PL**.75)*(PK**.25) =G= 100*PY

Infor given in value and  quantity of output X = 100
$PROD value and quantity of L = 75
block value and qujantity of K = 25

slope of isoquant at reference point = 1
curvature of CES, s = 1

value of output
=  100

relative prices 
of inputs = 1

curvature of 
isoquant  σ = 1

quantity of L: 75

qu
an

tit
y 

of
 K

: 2
5

this info used 
by MPS/GE to 
construct cost 
function and 
pricing 

One example of the treatment of taxes (others will follow later, including
those with endogenous rates) is in the production block for X.

(from kyiv2-mps)

$PROD:X s:1
        O:PX  Q:(100/(1+TC))
        I:PL  Q:25   A:CONS T:TR
        I:PK  Q:75   A:CONS T:TR

The “A” field means “assign” the revenue from tax TR to the agent CONS. 
Read it as the statement “assign to agent CONS the revenue from tax
rate TR on inputs L and K”.

A utility function is also represented by a production block; that is, utility is
a good which is produced from commodity inputs (including possibly
factor inputs such as leisure).  

Here is the utility function (W for welfare), in which utility (good PW) is
produced from inputs of X and Y.  MPS/GE constructs the underlying
expenditure (cost) function.

$PROD:W s:1
        O:PW   Q:200
        I:PX   Q:100
        I:PY   Q:100

A consumer’s income constraint is also represented by a “block” in this
case called a demand  block.  

In what follows, the consumer demands the utility good PW (the “D” field),
and receives income from endowments (the “E” fields) of labor and
capital.

MPS/GE automatically handles tax revenue or subsidy payments in the
background, adding or subtracting them to the consumer's
endowment income.

$DEMAND:CONS
        D:PW   Q:200
        E:PL   Q:(100*LENDOW)
        E:PK   Q:100

MPS/GE also automatically looks after the market clearing conditions.  

When the MPS/GE compiler sees “PL” listed under “commodities” it will
construct a supply $ demand inequality.

(1) find all sources of PL supply (outputs or endowments).  
In our case, there is only one, in $DEMAND: CONS 

(2) find all sources of PL demand (inputs). In our case, there are in
two, $PROD: X and $PROD: Y

(3) MPS/GE applies Shepard’s lemma to all cost functions to get the
demands for PL in the PROD blocks.

(4) constructs the supply/demand inequality.

Example of how the Q and P fields are used to construct the underlying
cost function.

All constant returns CES and CET functions can be completely
characterized by a single point consisting of (1) input quantities (2) output
quantities (3) input and output prices, and (4) the elasticity of substitution
or transformation (there may be several levels of substitution elasticities).

MPS/GE constructs the underlying cost function from such a single
observation.  It is particularly important to specify the reference prices
correctly.  Consider the two production blocks (based on kyiv4-mps).

$PROD:X s:1
        O:PX    Q:100        A:CONS T:TX  
        I:PL    Q: 40   P:1  A:CONS T:TLX 
        I:PK    Q: 60        A:CONS T:TKX 

$PROD:X s:1
        O:PX    Q:100        A:CONS T:TX  
        I:PL    Q: 20   P:2  A:CONS T:TLX 
        I:PK    Q: 60        A:CONS T:TKX 

With taxes in the benchmark, it is important to divide values into price and
quantity components when all prices cannot be normalized to one.  

In the above example, both of these blocks satisfy adding up, but they
represent different technologies.

The Q and P fields are used only in the calibration.  The P:2 must be here
in kyiv4-mps, when the tax is initially TLX = 1;  

The Q and P field should not be changed in counter-factuals unless you
mean to actually change the technology.

Armington formulation - goods differentiated by country of origin

Another feature of real data that confronts modelers is that there is
generally two-way trade in any “good” in the data; that is, each good is
both imported and exported.  

This is often attributed to the fact that data is actually classified by
industries and any industry is actually an aggregation of many goods.  

Yet two-way trade or “cross hauling” will persist in the data even at an
extremely fine level of disaggregation.   

No research will ever get data that is free of cross hauling so the question
is what to do about it.  

         Production Sectors                    Consumer

Markets|  X1     X2       E       M        W     CONS
------------------------------------------------------
P1     | 150           -100      50     -100
P2     |         50     -25      75     -100
PL     |-100    -20                               120
PK     | -50    -30                                80
PW     |                                 200     -200
PFX    |                125    -125
-----------------------------------------------------

Each “good” (industry) is both imported and exported.  One way to handle
this is to simply net out the two way trade from the gross flows and
replace the data with net trade only.  

If we do this in the present model, we arrive exactly a model essentially
the same as kyiv6.  We can then proceed as before.

The other alternative in competitive models is to assume that the
domestic and foreign goods are not really identical.  This is the so-
called Armington assumption: domestic and foreign goods in the
same industrial classification are imperfect substitutes.  

$PROD:W  s:1  G1:ESUB  G2:ESUB
        O:PW    Q:200
        I:P1    Q: 50  G1:
        I:PF1   Q: 50  G1:
        I:P2    Q: 25  G2:
        I:PF2   Q: 75  G2:

The domestic (P1) and foreign good (PF1) in industry 1 are in a nest as
are the domestic and foreign goods in industry 2 and these nest have
the same elasticity of substitution: 

A domestic and foreign good in the same industry have the same
substitution elasticity regardless of industry, but this is by no means
necessary.

The composite industry 1 good and the composite industry 2 good are
then combined in an upper level nest.  

This is generally assumed to have a lower elasticity of substitution: two
industries are poor substitutes than a domestic and foreign good in
the same industry.  




